

Z400系列测试代码-附带说明

REM 创建一个要打印的页面，宽度为100mm，高度为190mm，（每mm为8个点，以下的坐标是以点为单位，原则上此条命令需要在第一行）

SIZE 100 mm, 190 mm

REM 设置打印方向，第一个参数表示出纸方向，第二个参数标识是否镜像

DIRECTION 0,0

REM 设置打印完毕定位到缝隙，（SET GAP OFF打印后不定位）

SET GAP ON

REM 设置打印后切纸（SET CUTTER OFF为打印后不切纸）

SET CUTTER 1

REM 设置是否使用碳带，对于此款机器，始终关闭碳带

SET RIBBON OFF

REM 设置打印速度为3英寸/秒，可以设置1 1.5 2 2.5 3 3.5 4 4.5 5英寸/秒

SPEED 3

REM 设置打印黑度，参数范围0-15，整数

DENSITY 15

REM 清除页面缓存，相当于将页面初始化为空白

CLS

REM 画黑块，第一个参数为x坐标，第二个参数为起点y坐标，第三个参数为横向黑块宽度，第四个参数为黑块纵向高度

REM 画线相当于画一个比较细的黑块

REM 图片左上角坐标为0, 0,即坐标原点

BAR 300,10,4,90

BAR 30,100,740,4

BAR 30,880,740,4

BAR 30,1500,740,4

REM 绘制文字，第一个参数为x坐标，第二个参数为起点y坐标
REM 第三个参数为字体，目前支持TSS16.BF2 TSS24.BF2 TSS32.BF2
REM 第四、五个参数为纵横方向放大倍数
TEXT 400,25,"TSS24.BF2",0,3,3,"上海浦东2"
TEXT 30,120,"TSS24.BF2",0,1,1,"发件人：张三 (电话 874236021)"
TEXT 30,150,"TSS24.BF2",0,1,1,"发件人地址：广东省 深圳市 福田区 思创路123号工业园1栋2楼"
TEXT 30,200,"TSS24.BF2",0,1,1,"收件人：李四 (电话 13899658435)"
TEXT 30,230,"TSS24.BF2",0,1,1,"收件人地址：上海市 浦东区 太仓路司务小区9栋1105室"

TEXT 30,700,"TSS16.BF2",0,1,1,"各类邮件禁寄、限寄的范围，除上述规定外，还应参阅“中华人民共和国海关对”
TEXT 30,720,"TSS16.BF2",0,1,1,"进出口邮递物品监管办法”和国家法令有关禁止和限制邮寄物品的规定，以及邮”
TEXT 30,740,"TSS16.BF2",0,1,1,"寄物品的规定，以及邮电部转发的各国（地区）邮政禁止和限制。”
TEXT 30,760,"TSS16.BF2",0,1,1,"寄件人承诺不含有法律规定的违禁物品。”

REM 打印一维条码，第一个参数为x坐标，第二个参数为起点y坐标
REM 第7、8两个参数相同，表示线条粗细，最后一个参数为条码内容
BARCODE 80,300,"128",90,2,0,4,4,"873456093465"
BARCODE 550,910,"128",50,2,0,2,2,"873456093465"

REM 画方框，第一个参数为x坐标，第二个参数为起点y坐标（方框的左上角点的坐标）
REM 第三个参数为x坐标，第四个参数为起点y坐标（方框的右下角点的坐标）
REM 第五个参数线宽，第六个参数是圆角的弧度
BOX 40, 500, 340, 650, 4, 20

TEXT 60,520,"TSS24.BF2",0,1,1,"寄件人签字："
TEXT 130,625,"TSS24.BF2",0,1,1,"2015-10-30 09:09"

TEXT 50,1000,"TSS32.BF2",0,2,3,"广东 ---- 上海浦东"

REM 画圆，前两个参数为坐标值，第三个圆半径，第四个线宽
CIRCLE 700, 1400, 50, 6

TEXT 670,1370,"TSS24.BF2",0,3,3,"碎"

REM 打印二维条码QR CODE，前两个参数为坐标值，最后一个为二维码内容
QR CODE 620,620, H, 4, A, 0, M1, S0, "www.qrprt.com www.qrprt.com
www.qrprt.com"

REM 打印二维条码DATA MATRIX，前两个参数为坐标值，最后一个为二维码内容
DMATRIX 100,1300,400,400,"www.qrprt.com"

REM 将以上绘制的图形打印出来
PRINT 1,1

Z400系列TSPL指令集

1.打印机初始化

DIRECTION 0,0

SET GAP ON

SET CUTTER {0, 参数说明: 切刀是否启用, 参数: 0/1,默认参数: 0}

SPEED {0, 参数说明: 打印速度, 参数: 1-8, 默认参数: 8}

DENSITY {0, 参数说明: 打印浓度, 参数: 1-8, 默认参数: 7}

CLS

2.设置打印页大小 必须放在设置打印开始排版之前

SIZE {0, 参数说明: 打印宽度, 以实际用纸真实纸面大小设置} mm, {0, 参数说明: 打印高度, 以实际用纸真实纸面高度设置, 注意收缩2mm左右} mm

3.执行打印指令

PRINT 1,1

4.文本封装

TEXT {0},{1},{2},0,{3},{4},B1,\"{5}\"

TEXT {0},{1},{2},0,{3},{4},\"{5}\"

5.画文本框

BOX {0},{1},{2},{3},{4},{5}

6.线

BAR {0},{1},{2},{3}

LINE {0},{1},{2},{3},{M1-9} 最后M1、M2-M9参数可丢

7.一维码

BARCODE {0},{1},\"{2}\",{3},0,0,{4},{4},\"{5}\"

8.二维码

QRCODE {0},{1},{2},{3},A,{4},M2,S7,\"{5}\"

9.图片

BITMAP {0},{1},{2},{3},{4},

10. 查询打印机状态

READSTA

字体名称	字体大小 (宽*高) 单位 (像素)	字体大小 (宽*高) 单位 mm	字符集	
TSS12. BF2	12*12	1.5*1.5	GBK 中文	
TSS16. BF2	16*16	2*2	GBK 中文	
TSS24. BF2	24*24	3*3	GBK 中文	
TSS32. BF2	32*32	4*4	GBK 中文	
1. EFT	8*12	1*1.5	英文	兼容斑马
2. EFT	10*16	1.2*2	英文	兼容斑马
3. EFT	12*20	1.5*2.5	英文	兼容斑马
4. EFT	14*24	1.8*3	英文	兼容斑马
5. EFT	32*48	4*6	英文	兼容斑马
A. FNT	5*9	0.6*1.1	英文	兼容斑马
B. FNT	7*11	0.9*1.4	英文	兼容斑马
C. FNT	10*18	1.2*2.2	英文	兼容斑马
D. FNT	10*18	1.2*2.2	英文	兼容斑马
E. FNT	15*28	1.9*3.5	英文	兼容斑马
F. FNT	13*26	1.7*3.2	英文	兼容斑马
G. FNT	40*60	5*7.5	英文	兼容斑马
H. FNT	13*21	1.7*2.6	英文	兼容斑马